Appendix A

```
TITLE: Van der Linden (2007) model , B-GLIRT formulation
 DATA:
 FILE IS data.dat;

 VARIABLE:
 NAMES ARE y1-y10 t1-t10;
USEVARIABLES ARE y1-y10 t1-t10;
CATEGORICAL ARE y1-y10;

 ANALYSIS:
ALGORITHM=INTEGRATION;
ALGORITHM=ODLL;
MODEL:
abil by y1-y10*;                          ! link responses to ability factor
speed by t1-t10* (phi1-phi10);            ! link response times to speed factor

abil by t1-t10* (load1-load10);           ! link response times to the ability factor, we
                                          ! constrain these below
abil with speed@0;                        ! no covariance between the factors

[y1$1-y10$1] (bi1-bi10);            ! item difficulties
[t1-t10] (icept1-icept10);                ! time intensity parameters

abil@1;                                   ! fix the variance of the ability factor to 1
speed* (varT);                            ! define the variance of the speed factor which is
                                          ! constraint below
MODEL CONSTRAINT:
NEW(rho);                                 ! introduce a new parameter rho (which is rho_1 on
                                          ! Equation 9)
varT = 1- rho*rho;                        ! constraining the variance of the speed factor to put
                                          ! the parameters on the same scale as the
                                          ! hierarchical model


load1=-rho*phi1;                          ! specifying f(θ_p; ρ) for each item
load2=-rho*phi2;
load3=-rho*phi3;
load4=-rho*phi4;
load5=-rho*phi5;
load6=-rho*phi6;
load7=-rho*phi7;
load8=-rho*phi8;
load9=-rho*phi9;
load10=-rho*phi10;
```

Appendix B

```
TITLE: Thissen (1983) model
 DATA:
 FILE IS data.dat;

 VARIABLE:
 NAMES ARE y1-y10 t1-t10;
 USEVARIABLES ARE y1-y10 t1-t10;
CATEGORICAL ARE y1-y10;

 ANALYSIS:
ALGORITHM=INTEGRATION;
ALGORITHM=ODLL;
MODEL:
abil by y1-y10* (ai1-ai10);                ! link responses to ability factor
speed by t1-t10*;                          ! link response times to speed factor

abil by t1-t10* (load1-load10);            ! link response times to the ability factor, we
                                           ! constrain these below

abil with speed@0;                         ! no covariance between the factors

[y1$1-y10$1] (bi1-bi10);                   ! item difficulties
[t1-t10] (icept1-icept10);                 ! time intensity parameters

abil@1;                                    ! fix the variance of the speed and ability factors
speed@1;                                   ! to equal 1

MODEL CONSTRAINT:
NEW(rho);                                  ! introduce a new parameter rho (which is rho_1 on
                                           ! Equation 12)

load1=-rho*ai1;                            ! specifying f(θ_p; ρ) for each item
load2=-rho*ai2;
load3=-rho*ai3;
load4=-rho*ai4;
load5=-rho*ai5;
load6=-rho*ai6;
load7=-rho*ai7;
load8=-rho*ai8;
load9=-rho*ai9;
load10=-rho*ai10;
```

Appendix C

```
TITLE:    The IRTtree model
DATA:
FILE IS data.dat;

VARIABLE:
NAMES ARE y1-y10 t1-t10;
CATEGORICAL ARE y1-y10;

ANALYSIS:
type=RANDOM;

MODEL:
 abil by y1-y10*;                    ! link responses to ability factor
speed by t1-t10* ;                   ! link response times to speed factor

abil by t1-t10* (d) ;                ! link response times to the ability factor, this
                                     ! is rho_1 in Equation 13

speed with abil@0;                   ! no covariance between the factors

 abil@1;                             ! fix the variance of the speed and ability factors
 speed@1;                            ! to equal 1


quad | abil xwith speed;             ! define the interaction between speed and ability

t1 on quad (c);                      ! the interaction effect for each item, this will result
t2 on quad (c);                      ! in rho_2 in Equation 13
t3 on quad (c);
t4 on quad (c) ;
t5 on quad (c);
t6 on quad (c);
t7 on quad (c);
t8 on quad (c);
t9 on quad (c);
t10 on quad(c);
```

Appendix D

```
TITLE:    B-GLIRT Distance-Difficulty model for dichotomous data
DATA:
FILE IS data.dat;
VARIABLE:        NAMES ARE y1-y10 t1-t10;
CATEGORICAL ARE y1-y10;

ANALYSIS:
ALGORITHM=ODLL;
ALGORITHM=INTEGRATION;
type=RANDOM;
MODEL:
abil by y1-y10*1 (a1-a10);                    ! link responses to ability factor
speed by t1-t10*;                             ! link response times to speed factor

abil by t1-t10 (cross1-cross10);              ! link response times to the ability factor, we
                                              ! constrain these below

[y1$1-y10$1] (bi1-bi10);              ! item difficulties
[t1-t10] (icept1-icept10);                    ! time intensity parameters

abil@1;                                       ! fix the variance of the speed and ability factors
speed@1;                                      ! to equal 1

speed with abil@0;                            ! no covariance between the factors

f2sq | abil xwith abil;                       ! define the squared ability factor, θ^2

t1 on f2sq (qload1);                          ! the θ^2 effect for each item, this will be constrained
                                             !below
t2 on f2sq (qload2);
t3 on f2sq (qload3);
t4 on f2sq (qload4);
t5 on f2sq (qload5);
t6 on f2sq (qload6);
t7 on f2sq (qload7);
t8 on f2sq (qload8);
t9 on f2sq (qload9);
t10 on f2sq (qload10);

MODEL CONSTRAINT:
new(rho);                                     ! define the cross-relation parameter

cross1 = -2*rho*a1*bi1;                       ! this will give rho_1^star from Equation 17 subject
                                             to
cross2 = -2*rho*a2*bi2;                       ! Equation 15 as we have dichotomous data.
cross3 = -2*rho*a3*bi3;               ! In addition, we use -2 instead of 2 as Mplus uses
cross4 = -2*rho*a4*bi4;               ! the parameterization αθ-β instead of αθ+β
cross5 = -2*rho*a5*bi5;
cross6 = -2*rho*a6*bi6;
```

```
cross7 = -2*rho*a7*bi7;
cross8 = -2*rho*a8*bi8;
cross9 = -2*rho*a9*bi9;
cross10 = -2*rho*a10*bi10;

qload1 = rho * a1^2;                    ! this will give rho_2^star from Equation 17
qload2 = rho * a2^2;
qload3 = rho * a3^2;
qload4 = rho * a4^2;
qload5 = rho * a5^2;
qload6 = rho * a6^2;
qload7 = rho * a7^2;
qload8 = rho * a8^2;
qload9 = rho * a9^2;
qload10 = rho * a10^2;
```

Appendix E

```
TITLE:    B-GLIRT Distance-Difficulty model for Likert scale data
DATA:     FILE IS data.dat;

VARIABLE:
NAMES ARE y1-y10 t1-t10;
CATEGORICAL ARE y1-y10;

ANALYSIS:
ALGORITHM=ODLL;
ALGORITHM=INTEGRATION;
type=RANDOM;
MODEL:
ability by y1-y10*1 (a1-a10);          ! link responses to ability factor
speed by t1-t10*;                      ! link response times to speed factor

ability by t1-t10 (cross1-cross10);    ! link response times to the ability factor, we
                                       ! constrain these below

 [y1$2-y1$3] (b1_2-b1_3);              ! label the middle two threshold parameters
 [y2$2-y2$3] (b2_2-b2_3);              ! to enable constraining them below
 [y3$2-y3$3] (b3_2-b3_3);
 [y4$2-y4$3] (b4_2-b4_3);
 [y5$2-y5$3] (b5_2-b5_3);
 [y6$2-y6$3] (b6_2-b6_3);
 [y7$2-y7$3] (b7_2-b7_3);
 [y8$2-y8$3] (b8_2-b8_3);
 [y9$2-y9$3] (b9_2-b9_3);
 [y10$2-y10$3] (b10_2-b10_3);

[t1-t10] (icept1-icept10);             ! the time intensity parameters


ability@1;                             ! fix the variance of the speed and ability factors
speed@1;                               ! to equal 1

speed with abil@0;                     ! no covariance between the factors


f2sq | abil xwith abil;                ! define the squared ability factor, θ^2

t1 on f2sq (qload1);                   ! the θ^2 effect for each item, this will be constrained
t2 on f2sq (qload2);                   !below
t3 on f2sq (qload3);
t4 on f2sq (qload4);
t5 on f2sq (qload5);
t6 on f2sq (qload6);
t7 on f2sq (qload7);
t8 on f2sq (qload8);
t9 on f2sq (qload9);
```

```
t10 on f2sq (qload10);

MODEL CONSTRAINT:
new(rho);                                   ! define the cross-relation parameter

cross1 = 2*rho*a1*((b1_2+b1_3)/2);          ! this will give rho_1^star from Equation 17
cross2 = 2*rho*a2*((b2_2+b2_3)/2);          ! subject to Equation 18 as we have polytomous
cross3 = 2*rho*a3*((b3_2+b3_3)/2);          ! items
cross4 = 2*rho*a4*((b4_2+b4_3)/2);
cross5 = 2*rho*a5*((b5_2+b5_3)/2);
cross6 = 2*rho*a6*((b6_2+b6_3)/2);
cross7 = 2*rho*a7*((b7_2+b7_3)/2);
cross8 = 2*rho*a8*((b8_2+b8_3)/2);
cross9 = 2*rho*a9*((b9_2+b9_3)/2);
cross10 = 2*rho*a10*((b10_2+b10_3)/2);

qload1 = rho * a1^2;                        ! this will give rho_2^star from Equation 17
qload2 = rho * a2^2;
qload3 = rho * a3^2;
qload4 = rho * a4^2;
qload5 = rho * a5^2;
qload6 = rho * a6^2;
qload7 = rho * a7^2;
qload8 = rho * a8^2;
qload9 = rho * a9^2;
qload10 = rho * a10^2;
```